1. IDENTIFICATION
   1.1 Digital-8-20-U-Sym
   1.2 Character String Typeout
   1.3 October 22, 1965
2. ABSTRACT
   A basic subroutine to type messages stored internally as a string of coded characters. All
   ASR-33 characters are legal.

3. REQUIREMENTS

3.1 Storage
   This subroutine uses 59 (decimal) core memory locations.

3.3 EQUIPMENT
   Basic PDP-8

4. USAGE

4.1 Loading
   This subroutine may be placed in core through the use of the Binary Loader, which is com­
   pletely described in Digital-8-2-Rim. The library tape supplied is symbolic.

4.2 Calling Sequence
   Call with a JMS with the starting address of the character string in the AC. Return will be
   to the instruction following the calling JMS.

5. RESTRICTIONS (Not Applicable)

6. DESCRIPTION

6.1 Discussion
   The ASCII character set breaks naturally into two major groupings: characters represented
   by codes 240 through 277; and characters represented by codes 301 through 337. Characters with these
   codes may readily be handled by representing them internally as stripped 6-bit codes. See Digi­
   tal-8-18-U-Sym and Digital-8-19-U-Sym for a complete discussion of how this is done.

   The following are special characters:

<table>
<thead>
<tr>
<th>Character</th>
<th>Code</th>
</tr>
</thead>
<tbody>
<tr>
<td>EOT</td>
<td>204</td>
</tr>
<tr>
<td>WRU</td>
<td>205</td>
</tr>
<tr>
<td>RU</td>
<td>206</td>
</tr>
<tr>
<td>BELL</td>
<td>207</td>
</tr>
<tr>
<td>Line Feed</td>
<td>212</td>
</tr>
<tr>
<td>Return</td>
<td>215</td>
</tr>
<tr>
<td>@</td>
<td>300</td>
</tr>
<tr>
<td>ACK</td>
<td>374</td>
</tr>
<tr>
<td>ALT MODE</td>
<td>375</td>
</tr>
<tr>
<td>RUBOUT</td>
<td>377</td>
</tr>
</tbody>
</table>

   These special characters are represented by codes which conflict with the groupings from 240 to 277 and
   301 to 337. Consequently when these characters must be output, they are treated as exceptions and
developed by special methods as described in Digital-8-18-U-Sym and Digital-8-19-U-Sym. Neither of these programs permits the development of all the codes listed above. This program does.

7. METHODS

7.1 Discussion

Internally characters are represented as 6-bit stripped characters and are packed two to a word. The stripped character 00 is used to indicate that the following character is a special character. For example, @ may be developed by packing 0000.

Since the appearance of 00 indicates that the next 6-bit group is to receive special treatment, 64 special characters are possible. This is many more than necessary to accommodate the ten special characters listed above that are required for ASCII typeout. The 6-bit group 000001 is therefore used to indicate the end of a given character string since it is not needed for regular ASCII output.

The method is straightforward. The first message word is picked up and the two trimmed codes masked out. Two jumps to the subroutine tagged TSCC2 are made in order to type the two characters. TSCC2 tests first to determine if the special character flag is set indicating that the current character is special. If so, a JMP to TYPSP is executed. If not, a test is made to see if the current code is 00. If so, the special character flag is set but no typeout ensues. If not, a regular character is being processed and is typed.

The TYPSP section of coding processes special characters. The special characters may be classified as:

<table>
<thead>
<tr>
<th>Special Character</th>
<th>Comments</th>
</tr>
</thead>
<tbody>
<tr>
<td>300</td>
<td>Logically the lowest element of extended group 301 through 337.</td>
</tr>
<tr>
<td>374, 375, 377</td>
<td>Least significant two digits similar to those in group 240 to 277.</td>
</tr>
<tr>
<td>204, 205, 206</td>
<td>Least significant two digits similar to those in group 207, 212, 215</td>
</tr>
<tr>
<td>207, 212, 215</td>
<td>301 through 337.</td>
</tr>
</tbody>
</table>

In order to develop the correct output, TYPSP changes the SPA command in SWITCH to a SMA command for all special characters but 300.

8. FORMAT

8.4 Miscellaneous

Refer to Digital-8-18-U-Sym and Digital-8-19-U-Sym for further format and code description.

9. EXECUTION TIME (Not applicable)
10. PROGRAM

10.4 Program Listing

/DIGITAL 8-20-U
/CHARACTER STRING TYPE-OUT
/CALL WITH STRING ADDRESS IN
/C(AC); ALL CODES MAY BE DEVELOPED
/RETURN FOLLOWING THE JMS

0200 0000 TYPSTG, @
0201 3262 DCA TEMQ /STORE INITIAL ADDRESS
0202 3264 DCA FLAG /CLEAR FLAG
0203 1662 TSCC1, TAD I TEMQ /PICK UP DATA
0204 7012 RTR /ROTATE 6 BITS RIGHT
0205 7012 RTR
0206 7012 RTR
0207 4214 JMS TSCC2 /TYPE FIRST CHARACTER
0210 1662 TAD I TEMQ /PICK UP DATA
0211 4214 JMS TSCC2 /TYPE SECOND CHARACTER
0212 2262 ISZ TEMQ /INCREMENT STORAGE ADDRESS
0213 5203 JMP TSCC1 /GO BACK FOR MORE

0214 0000 TSCC2, @
0215 0265 AND K77 /MASK OFF 6 BITS
0216 3263 DCA TEMR /SAVE CHARACTER
0217 1264 TAD FLAG /TEST "SPECIAL" FLAG
0220 7640 SZA CLA
0221 5231 JMP TYPSP /SET: TYPE SPECIAL
0222 1263 TAD TEMR /NO: REGULAR CHARACTER
0223 7450 SNA /IS IT ZERO?
0224 5227 JMP .+3 /YES: SET FLAG
0225 4250 TYPAT, JMS PRINT /NO: PRINT IT
0226 5614 JMP I TSCC2 /RETURN
0227 2264 ISZ FLAG /SET "SPECIAL" FLAG
0230 5614 JMP I TSCC2 /EXIT

0231 3264 TYPSP, DCA FLAG /CLEAR "SPECIAL" FLAG
0232 1263 TAD TEMR /TEST FOR "@"
0233 7041 CIA
0234 7450 SNA
0235 5225 JMP TYPAT /@: TYPE "@"
0236 7001 IAC /TEST FOR 01
0237 7650 SNA CLA
0240 5600 JMP I TYPSTG /YES: EXIT CODE
0241 1271 TAD SKIPMA /ALTER INSTRUCTION
0242 3252 DCA SWITCH /TO BE "SMA"
0243 1263 TAD TEMR /TYPE CHARACTER
0244 4250 JMS PRINT
0245 1272 TAD SKIPPAA /ALTER INSTRUCTION
0246 3252 DCA SWITCH /TO BE "SPA"
0247 5614 JMP I TSCC2 /RETURN

0250 0000 PRINT, @
0251 1266 TAD M40 /COMPARE WITH 40
0252 7510 SWITCH, SPA /OR SMA FOR SPECIAL CODES
Digital-S-20-U-Sym
Page 4

\[ \begin{align*}
\omega 253 &: 1267 & \text{TAD C100} \\
\omega 254 &: 1270 & \text{TAD C240} \\
\omega 255 &: 6446 & \text{TLS} \\
\omega 256 &: 6041 & \text{TSF} \\
\omega 257 &: 5256 & \text{JMP -1} \\
\omega 261 &: 7200 & \text{CLA} \\
\omega 261 &: 5650 & \text{JMP I PRINT}
\end{align*} \]

\text{/CONSTANTS AND TEMPORARY REGISTERS}

\[ \begin{align*}
\omega 262 &: 0000 & \text{TEMQ, 0} & \text{/CONTAINS STRING ADDRESS} \\
\omega 263 &: 0000 & \text{TEMR, 0} & \text{/CONTAINS 6 BIT CHARACTER} \\
\omega 264 &: 0000 & \text{FLAG, 0} & \text{/"SPECIAL" FLAG} \\
\omega 265 &: 0077 & \text{K77, 77} \\
\omega 266 &: 7740 & \text{M40, -40} \\
\omega 267 &: 1100 & \text{C100, 100} \\
\omega 270 &: 0240 & \text{C240, 240} \\
\omega 271 &: 7500 & \text{SKIPMA, SMA} \\
\omega 272 &: 7510 & \text{SKIPPA, SPA}
\end{align*} \]

\[ \begin{align*}
\text{C100} &: \omega 267 \\
\text{C240} &: \omega 270 \\
\text{FLAG} &: \omega 264 \\
\text{K77} &: \omega 265 \\
\text{M40} &: \omega 266 \\
\text{PRINT} &: \omega 250 \\
\text{SKIPMA} &: \omega 271 \\
\text{SKIPPA} &: \omega 272 \\
\text{SWITCH} &: \omega 252 \\
\text{TEMQ} &: \omega 262 \\
\text{TEMR} &: \omega 263 \\
\text{TSCC1} &: \omega 203 \\
\text{TSCC2} &: \omega 214 \\
\text{TYPAT} &: \omega 225 \\
\text{TYPSP} &: \omega 231 \\
\text{TYPSTG} &: \omega 200
\end{align*} \]

11. DIAGRAMS (Not Applicable)

12. REFERENCES

12.1 Other Library Programs
Digital-8-18-U-Sym and Digital-8-19-U-Sym